AmirHosein Sadeghimanesh
2022 July

Consider the example in Example 4.7 of the paper. Here we do the calculations that simplify the system of steady state equations to a single univariate polynomial.

The initial system of steady state equations is the following.

```
> F := [ ( 4 + 8*x[4]^4  / ( 256 + x[4]^4 ) ) * x[3]^(-1/2) - (1/2)
  *x[1]^(1/2),
  x[1] - k[1]*x[2]^(1/2),
  2*x[2] -k[2]*x[3],
  3*x[3] - k[3]*x[4]^(3/4) ]:
  < seq(f, f in F) >;
```

$$\begin{bmatrix} \dfrac{4+\dfrac{8\,x_4^4}{x_4^4+256}}{\sqrt{x_3}}-\dfrac{\sqrt{x_1}}{2} \\ x_1-k_1\sqrt{x_2} \\ -k_2\,x_3+2\,x_2 \\ 3\,x_3-k_3\,x_4^{3/4} \end{bmatrix} \quad (1)$$

Solving the 4th equation to find x_3 as a function of x_4.

```
> solve( F[4], x[3] );
```

$$\frac{k_3\,x_4^{3/4}}{3} \quad (2)$$

Now solving the 3rd equation to find x_2 as a function of x_3 and the substituting the previous result to write x_2 as a function of x_4.

```
> solve( F[3], x[2] );
```

$$\frac{k_2\,x_3}{2} \quad (3)$$

```
> eval( (3), x[3] = (2) );
```

$$\frac{k_2\,k_3\,x_4^{3/4}}{6} \quad (4)$$

Now solving the 2nd equation to write x_1 as a function of x_2 and then substituting the result for x_2 to write x_1 as a function of x_4

```
> solve( F[2], x[1] );
```

$$k_1 \sqrt{x_2} \tag{5}$$

```
> eval( (5), x[2] = (4) );
```

$$\frac{k_1 \sqrt{6} \sqrt{k_2 k_3 x_4^{3/4}}}{6} \tag{6}$$

Now substituting results of x_1 and x_3 in the 1st equation to get an equation involving only x_4.

```
> eval( F[1], [ x[1] = (6), x[3] = (2) ] );
```

$$\frac{\left(4 + \frac{8 x_4^4}{x_4^4 + 256}\right) \sqrt{3}}{\sqrt{k_3 x_4^{3/4}}} - \frac{\sqrt{6} \sqrt{k_1 \sqrt{6} \sqrt{k_2 k_3 x_4^{3/4}}}}{12} \tag{7}$$

If we define y = x_4^{1/16}, then x_4=y^{16} and we can rewrite the above equation free of radicals.

```
> eval( %, x[4] = y^16 );
```

$$\frac{\left(4 + \frac{8 y^{64}}{y^{64} + 256}\right) \sqrt{3}}{\sqrt{k_3 \left(y^{16}\right)^{3/4}}} - \frac{\sqrt{6} \sqrt{k_1 \sqrt{6} \sqrt{k_2 k_3 \left(y^{16}\right)^{3/4}}}}{12} \tag{8}$$

```
> simplify( %, assume = positive );
```

$$-\frac{\sqrt{3} \left(y^9 2^{3/4} 3^{1/4} \sqrt{k_1}\, k_2^{1/4} \left(y^{64} + 256\right) k_3^{3/4} - 144 y^{64} - 12288\right)}{12 \sqrt{k_3} \left(y^{64} + 256\right) y^6} \tag{9}$$

Since the denominator is strictly positive for positive values of k_3 and x_4, we can simply only keep the numerator when studing this equation for its roots.

```
> numer( % );
```

$$-\sqrt{3} \left(2^{3/4} \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} y^{73} - 144 y^{64} + 256\, 2^{3/4} y^9 \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} - 12288\right) \tag{10}$$

```
> collect( 2^(3/4)*sqrt(k[1])*3^(1/4)*k[2]^(1/4)*k[3]^(3/4)*y^73 -
  144*y^64 + 256*2^(3/4)*y^9*sqrt(k[1])*3^(1/4)*k[2]^(1/4)*k[3]^
  (3/4) - 12288, y );
```

$$2^{3/4} \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} y^{73} - 144 y^{64} + 256\, 2^{3/4} y^9 \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} - 12288 \tag{11}$$

```
> g := %;
```

$$g := 2^{3/4} \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} y^{73} - 144 y^{64} + 256\, 2^{3/4} y^9 \sqrt{k_1}\, 3^{1/4} k_2^{1/4} k_3^{3/4} - 12288 \tag{12}$$

This is the univariate polynomial given in the text.

To further simplify the representation of the coefficients note that

```
> ( 2^(3/4)*sqrt(k[1])*3^(1/4)*k[2]^(1/4)*k[3]^(3/4) )^4;
```

$$24\, k_1^2\, k_2\, k_3^3 \qquad \textbf{(13)}$$

So instead of 2^(3/4)*sqrt(k[1])*3^(1/4)*k[2]^(1/4)*k[3]^(3/4) we can write (24*k[1]^2*k[2]*k[3]^3)^(1/4) .

End of the file.